

Module 6

Holding & Matching Files

In this module you will learn:

- How to hold data for further processing
- How to view the master file description of the hold file
- How to merge data from multiple files

Supporting Documents/Files:

- FOCUS Documentation
- CIRS Data Element Dictionary
- CIRS Common Library

Overview

The topics in this module will describe how to gather data for your reports using the HOLD command and the MATCH command. The HOLD command allows you to gather data and save the result to a file for reuse. The MATCH command allows you to merge two or more data files, and specify which records will be merged and which will be screened out, based on the merge phrase. The resulting merged fields are placed in a new data file—a HOLD file.

Holding Files

FOCUS provides several options to save and reuse your report output. This module will discuss only the HOLD file. Refer to your FOCUS documentation for additional options.

The HOLD file type enables you to save report data in a file, rather than display the report on the screen. In prior training, you learned to use this command to place data in a permanent hold file for downloading purposes and/or with the extract process. This module will discuss creating hold files for further processing in multi-step reports.

Hold File Syntax

Hold files allow you to retrieve and process data, then save the results for further processing. The syntax is:

```
EX filename
TABLE FILE filename
display command
sort phrase
screening criteria
ON TABLE HOLD AS 'newfile'
END
```

- Only fields requested in a verb or sort line will be available for further processing. Fields used for screening will not be available.
- A HOLD file is available for the entire FOCUS session. If desired, you can save the output to one of your permanent hold files.
- If a format type is not specified, a FOCUS file format is created along with a masterfile description (MFD).
- To view the MFD for a HOLD file, use the command:
? HOLD filename
- To display the contents of a HOLD file, use the PRINT command with the wildcard character (*).

Hold Files - Example

The report below counts the number of full-time and part-time Faculty. For the purposes of this request, a Faculty employee is anyone in collective bargaining id R03. Full-time employees are those whose total time base of all R03 positions is greater than or equal to .997. Each step required for this report will be discussed below.

Report Generated (final):

GROUP	AC:SSA COUNT
-----	-----
FULL-TIME	358
PART-TIME	152
TOTAL	510

Step 1: Gather your data from the AC file.

```
-* TOTAL FTE BY SSA  
EX AC  
TABLE FILE AC  
SUM AC:FTE  
BY AC:SSA  
IF AC:CBID EQ R03  
ON TABLE HOLD AS 'PERMSML'  
END
```

Step 2: Verify the data in the hold file.

```
-* CONTENTS OF PERMSML FILE  
TABLE FILE PERMSML  
PRINT *  
END
```

AC:SSA	AC:FTE
-----	-----
000-03-1111	.333
001-10-1111	.512
001-24-1111	.500
001-28-1111	.200
001-34-1111	1.000
001-36-1111	1.000
001-37-1111	1.000
001-42-1111	.400

Hold File Example – Con’t

Step 3: Check the Master File Description of your hold file.

```
? HOLD PERMSML

DEFINITION OF HOLD FILE: PERMSML
FIELDNAME                ALIAS                FORMAT

AC:SSA                    E01                A11
AC:FTE                    E02                P9.3
```

Step 4: Develop the report from the hold file.

```
DEFINE FILE PERMSML
GROUP/A10 = IF AC:FTE GE .997 THEN
            `FULL-TIME' ELSE `PART-TIME';
END
TABLE FILE PERMSML
COUNT AC:SSA
BY GROUP
ON TABLE COLUMN-TOTAL
END
```

Step 5: Put it all together in a single report request.

After you have tested all steps of your request, assemble the individual requests into a single report request using the COPY command and remove the record limits. This is the version you will keep in your library. All requests created during development can be deleted, if desired.

```
-* COUNT OF PART-TIME AND FULL-TIME R03 EMPLOYEES
EX AC
TABLE FILE AC
SUM AC:FTE
BY AC:SSA
IF AC:CBID EQ R03
ON TABLE HOLD AS `PERMSML'
END
DEFINE FILE PERMSML
GROUP/A10 = IF AC:FTE GE .997 THEN
            `FULL-TIME' ELSE `PART-TIME';
END
TABLE FILE PERMSML
COUNT AC:SSA
BY GROUP
ON TABLE COLUMN-TOTAL
END
```


Matching Files

You can merge two or more data files, and specify which records will be merged and which will be screened out, using the MATCH command. The command creates a new data file—a HOLD file—that contains the merged fields from the selected records. You can report from the new data file and use it as you would any other HOLD file.

You select which records are merged into the new data file and which records are excluded by specifying sort fields in the MATCH command. You specify one set of sort fields—using the BY phrase—for the first data file, and a second set of sort fields for the second data file. The MATCH command compares all sort fields that have been specified in common for both data files, and then merges into the new HOLD file all records from the first data file whose sort values match those in the second data file.

In addition to merging data file records that share values, you can merge records based on other relationships. For example, you can merge all records whose values do not match—that is, all records in each data file whose sort values are not matched in the other data file.

The MATCH command would be used in the following situations:

- If you need data from more than one data file and the files are not joined.
- If you need to compare or contrast the data file records.
- If you need data from multiple segments of a multi-path file (e.g., the Pay Scales file).
- If you need data from a parent segment that has ‘missing descendants’ (e.g., the payment segment of the Payment Data file).


Match Files - Syntax

The syntax of the MATCH command is similar to that of the TABLE command:

DEFINES	EX filename
OLD	MATCH FILE filename display command BY filename screening RUN
NEW	FILE filename display command BY filename screening AFTER MATCH HOLD AS 'permfile' merge phrase END
REPORT	TABLE FILE permfile report lines END


- All standard defines are executed before the match request.
- The first data file is considered the OLD file. The second file is considered the NEW file. The data in the OLD and NEW file will be compared, or combined, based on the merge phrase.
- At least one BY field must be specified for each file used in the MATCH request.
- Up to six sets of data can be merged in one MATCH request. A RUN command must follow each AFTER MATCH phrase (except for the last one). The END command must follow the final AFTER MATCH phrase.
- The resulting merged data is placed in a single-segment HOLD file. The HOLD file may be matched to additional files, if needed.
- The contents of the HOLD file can be displayed using the PRINT command with the wildcard character (*). To view the MFD, use the command: **? HOLD.**
- The commands ACROSS, COMPUTE and BY HIGHEST are not permitted in a MATCH request. You can, however, use the DEFINE command.

Match Files – Merge Phrase




OLD-OR-NEW

Combine the two files. This is the default if no merge phrase is specified.




OLD-AND-NEW

Keep records only if they appear in both reports.




OLD-NOR-NEW

Keep ALL records from the OLD file that do not have a matching record in the NEW file, and all records from the NEW file that do not have a matching record in the OLD file.




OLD

Keep ALL records from the OLD report and add in only records that match from the NEW report.




NEW

Keep ALL records from the NEW report and add in only records that match from the OLD report.



OLD-NOT-NEW

Keep records that appear only in the OLD report.



NEW-NOT-OLD

Keep records that appear only in the NEW report.

Match Files - Processing

The way the MATCH command merges data depends on the order in which the data files are named in the request, the BY fields, the display commands, and the merge phrases used. In general, processing is as follows:

1. The requested records are retrieved from the first data file named and are written to a temporary work area.
2. The requested records are retrieved from the second data file named and are written to a temporary work area.
3. The retrieved records from the first and second data files are compared based on the common sort fields specified in the merge phrase (for example, OLD-OR-NEW).
4. The merged results of the comparison are written to a temporary data file. If there are more MATCH operations, the system cycles through all data files named until END is encountered.
5. The final records are written to the HOLD file.

Match Files – Missing Data

Sometimes, your hold data file will contain missing data (null values). This commonly occurs when you match multiple occurrences in one file to a single occurrence in another file. For example, matching many payments for an employee in the PH file to a single ethnic code for the same employee in the AC file.

Solution #1

When possible, use the verb 'SUM' to get one record per employee in BOTH the old and new files.

Solution #2

Use defines to fill-in the missing data in the hold file. For an example, refer to common library member: FILLIN.

Match Files - Example

The report below identifies employees permanently separated effective January 1, 1998 or later.

Report Generated:

EMPLOYEES PERMANENTLY SEPARATED
SINCE 01/01/1998

EMPLOYEE	SSA	SEPARATION DATE	PSN SEQ
-----	---	-----	---
ARDVARK, ANNA	009-00-0000	06/04/1999	02
ANDCANDA, JOHN	444-00-0000	12/18/2001	01
BEAR, RICHARD	003-00-0000	08/01/2000	01
		08/01/2000	02
CAMEL, HILARY	555-00-0000	07/31/1998	01
DOG, MICHAEL	111-00-0000	08/01/2000	01

To identify these employees, a comparison between social security numbers in the Separated Current Status (SP) file and the Active Current Status (AC) file is required. Using the NEW-NOT-OLD match option, employees are selected only if they have a record in the SP file and no record in the AC file.

Note: The AC file is updated weekly, while the SP file is updated monthly. Therefore, it is possible for an employee to separate and be eliminated from the AC file, but to still have a record in the SP file. For maximum accuracy, this report should be run immediately after the SP file is updated at the beginning of the month.

Report Request Steps:

1. Identify the file(s) that contain the data needed.
2. Write a report request to gather data from each file to obtain data for verification. Be sure to include record limits for testing purposes.
3. Write a match request and sorting by the same fieldname in each file.
4. Write a report request to view and verify the contents of the hold file.
5. Write the report using data from the hold file.
6. Put it all together in a single report request.

Match Files – Example Con't

Step 1: Identify the file(s) that contain the data needed.

AC – has current positions as of close of business on prior Friday
SP – has separated positions from 1987 through the prior month

Step 2: Write report requests (e.g., PERMSEP1, PERMSEP2) to gather data from each file for verification purposes. Be sure to include record limits for testing purposes.

```
EX AC
TABLE FILE AC
PRINT AC:WNAME
BY AC:SSA
IF AC:SSA LE 025-00-0000
END
```

```
EX SP
TABLE FILE SP
PRINT SP:WNAME SP:PSNSEQ SP:EFFDATE
BY SP:SSA
IF SP:EFFDATE GE 01011998
IF SP:SSA LE 025-00-0000
END
```

Step 3: Write a match request (e.g., PERMSEP3). Be sure to sort by the same fieldname in each file.

```
EX AC
EX SP
-*
MATCH FILE AC
PRINT AC:WNAME
BY AC:SSA AS 'SSA'
IF AC:SSA LE 025-00-0000
RUN
-*
FILE SP
PRINT SP:WNAME SP:PSNSEQ SP:EFFDATE
BY SP:SSA AS 'SSA'
IF SP:EFFDATE GE 01011998
IF SP:SSA LE 025-00-0000
AFTER MATCH HOLD AS 'PERMSML'
NEW-NOT-OLD
END
```

Match Files – Example Con't

Step 4: Write a report request to view and verify the contents of the hold file (e.g, PERMSEP4).

```
TABLE FILE PERMSML
PRINT *
END
```

Step 5: Write a report using the hold file data (e.g., PERMSEP5).

```
TABLE FILE PERMSML
PRINT SP:EFFDATE AS 'SEPARATION, DATE'
      SP:PSNSEQ AS 'PSN, SEQ'
BY SP:WNAME AS 'EMPLOYEE'
BY SSA
END
```

Step 6: Put it all together in a single report request (e.g., PERMSEP).

After you have tested all steps of your match request, you can remove the record limits and assemble the individual report requests into a single report request using the COPY command. This is the version you would keep in your library. All requests created during development can be deleted, if desired.

```
-* PERMANENTLY SEPARATED EE'S SINCE 01/01/1998
EX AC
EX SP
-*
MATCH FILE AC
PRINT AC:WNAME
BY AC:SSA AS 'SSA'
RUN
-*
FILE SP
PRINT SP:WNAME SP:PSNSEQ SP:EFFDATE
BY SP:SSA AS 'SSA'
IF SP:EFFDATE GE 01011998
AFTER MATCH HOLD AS 'PERMSML'
NEW-NOT-OLD
END
-*
TABLE FILE PERMSML
PRINT SP:EFFDATE AS 'SEPARATION, DATE'
      SP:PSNSEQ AS 'PSN, SEQ'
BY SP:WNAME AS 'EMPLOYEE'
BY SSA
END
```


Match Files – Practice Con't

- Copy the common library report: TOTALPAY into your library. This report is a match between the payment and deduction segments of the PH file. It generates a report that displays the total earnings and state share (benefit) amounts for an employee. Modify the report per the instructions, using your own social security number and campus code.

Report Generated:

SOC-SEC-NUM	EMPLOYEE	PAY PERIOD	GROSS PAY	TOTAL BENEFITS*	TOTAL
111-22-3333	ANT, SUE	2002/05	\$3,500.00	\$1,122.00	4,622.00

Report Commands:

A portion of the common report request: TOTALPAY is shown below:

```

MATCH FILE PH
SUM PH:GROSSPAY PH:SSTAXWHS PH:MDTAXWHS PH:RETSTSHR
BY PH:SSA BY PH:WNAME BY PH:PAYPERIOD
IF PH:CAMPX EQ -
IF PH:SSA EQ -
IF PH:PAYPERIOD EQ -
RUN
-*
FILE PH
SUM HEALTH DENTAL VISION LIFEINS FLEXHEALTH FLEXDENTAL FLEXCASH
BY PH:SSA BY PH:WNAME BY PH:PAYPERIOD
IF PH:CAMPX EQ -
IF PH:SSA EQ -
IF PH:PAYPERIOD EQ -
AFTER MATCH HOLD AS PERMSML OLD-OR-NEW
END
-*****
-* THE LINES BELOW PRODUCE THE FINAL REPORT
-*****
DEFINE FILE PERMSML
TOTAL_BEN/P12.2M =
(PH:SSTAXWHS + PH:MDTAXWHS + PH:RETSTSHR + HEALTH + DENTAL +
VISION + LIFEINS + FLEXHEALTH + FLEXDENTAL + FLEXCASH);
END
-*
TABLE FILE PERMSML
HEADING CENTER
"TOTAL WAGES AND BENEFITS, BY EMPLOYEE"
BY PH:SSA AS 'SOC-SEC-NUM'
BY PH:WNAME AS 'EMPLOYEE'
PRINT PH:PAYPERIOD AS 'PAY, PERIOD'
PH:GROSSPAY AS 'GROSS PAY'
TOTAL_BEN AS 'TOTAL, BENEFITS*'
ON TABLE ROW-TOTAL
FOOTING
"*BENEFITS INCLUDE: STATE SHARE FOR SOCIAL SECURITY, MEDICARE,"
"RETIREMENT, HEALTH, DENTAL, VISION, LIFE INSURANCE & FLEX CASH"
END
    
```


What You Have Learned

In this module you learned how to:

- Hold data for further processing
- View the master file description of the hold file
- Merge data from multiple sources